HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS
Physics



INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁎. When you see ⁎ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL^{T}SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02^{T}THIS IS A*
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	*03^T ⊢ SAMPLE*
04 AVIEW	■ AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] ■ [•] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	*11^{T}R3=*
12 ARCL 03	■ [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	■ [RTN]	*14 RTN*

TABLE OF CONTENTS

* These programs require one additional memory module.

BLACK BODY THERMAL RADIATION

Bodies with finite temperatures emit thermal radiation. The higher the absolute temperature, the more thermal radiation emitted. Bodies which emit the maximum possible amount of energy at every wavelength for a specified temperature are said to be black bodies. While black bodies do not actually exist in nature, many surfaces may be assumed to be black for engineering considerations.



Figure 1.

Notes:

A half minute or more may be required to obtain $E_{b(0-\lambda)}$ or $E_{b(\lambda_1-\lambda_2)}$ since the integration is numerical.

Sources differ on values for constants. This could yield small discrepancies between published tables and program outputs.

Figure 1 is a representation of black body thermal emission as a function of wavelength. Note that as temperature increases, the area under the curves (total emissive power $E_{b(0-\infty)}$) increases. Also note that the wavelength of maximum emissive power λ_{max} shifts to the left as temperature increases.

This program calculates the wavelength of maximum emissive power for a given temperature, the temperature for which a given wavelength would be the wavelength of maximum emissive power, the total emissive power over all wavelengths, the emissive power at a particular wavelength, the emissive power from zero to a specified wavelength, and the emissive power between specified wavelengths.

Equations:

$$\lambda_{max} T_{\lambda_{max}} = c_3$$

$$E_{b(0-\infty)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

$$E_{b(0-\lambda)} = \int_0^\lambda E_{b\lambda} \, d\lambda$$

$$= 2\pi c_1 \sum_{k=1}^{\infty} -T/kc_2 \, e^{-\frac{kc_2}{T\lambda}} \left[\left(\frac{1}{\lambda}\right)^3 + \frac{3T}{\lambda^2 kc_2} + \frac{6}{\lambda}\left(\frac{T}{kc_2}\right)^2 + 6\left(\frac{T}{kc_2}\right)^3 \right]$$

$$E_{b(\lambda_1-\lambda_2)} = E_{b(0-\lambda_2)} - E_{b(0-\lambda_1)}$$

where:

λ_{max} is the wavelength of maximum emissivity in microns;

T is the absolute temperature in °R or K;

$E_{b(0-\infty)}$ is the total emissive power in Btu/hr-ft^2 or Watts/cm^2;

$E_{b\lambda}$ is the emissive power at λ in Btu/hr-ft^2-μm or Watts/cm^2-μm;

$E_{b(0-\lambda)}$ is the emissive power for wavelengths less than λ in Btu/hr-ft^2 or Watts/cm^2;

$E_{b(\lambda_1-\lambda_2)}$ is the emissive power for wavelengths between λ_1 and λ_2 in Btu/hr-ft^2 or Watts/cm^2.

$c_1 = 1.8887982 \times 10^7$ Btu-μm^4/hr-ft^2
$= 5.9544 \times 10^3$ Wμm^4/cm^2

$c_2 = 2.58984 \times 10^4$ μm-°R $= 1.4388 \times 10^4$ μm-K

$c_3 = 5.216 \times 10^3$ μm-°R $= 2.8978 \times 10^3$ μm-K

$\sigma = 1.713 \times 10^{-9}$ Btu/hr-ft^2-°R^4 $= 5.6693 \times 10^{-12}$ W/cm^2-K^4

$\sigma_{exp} = 1.731 \times 10^{-9}$ Btu/hr-ft^2-°R^4 $= 5.729 \times 10^{-12}$ W/cm^2-K^4

References: HP-67/97 Users' Library Program.

Example:

What percentage of the radiant output of a lamp is in the visible range (0.4 to 0.7 microns) if the filament of the lamp is assumed to be a black body at 2400K?

Keystrokes: (SIZE ⩾ 009)	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] BB [ALPHA]	UNITS?
SI [R/S]	TEMP?
2400 [R/S]	WAVELENGTH?
.4 [R/S]	SOLVE
[F]	WV LNTH 2?
.7 [R/S]	EbL-L=4.9679
[C]	EbTOT=188.094
[÷]	0.0264
100 [x]	2.6412

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode.		[USER]	
2.	Initialize program		[XEQ] BB	UNITS?
3.	Input code for desired units SI, or EN.	SI	[R/S]	TEMP?
		EN	(or) [R/S]	TEMP?
4.	Input temperature. If temperature is	Temp	[R/S]	WAVELENGTH?
	unknown, press [R/S].			
5.	Input first wavelength. If wavelength is	λ	[R/S]	SOLVE
	unknown, press [R/S].			
6.	Calculate any or all of the following:			
	λmax for a given temperature		[A]	WLMAX=
	T such that λ is λmax for T		[B]	TEMP=
	total emissive power at T		[C]	EbTOT=
	emissive power at T and λ		[D]	EbL=
	emissive power between zero and λ		[E]	Eb0-L=
	emissive power between λ_1 and λ_2		[F]	WV LNTH 2?
		λ_2	[R/S]	EbL-L=
7.	For a new case, go to step 2.			

Program Listings

Initialize and prompt for units

```
01*LBL "BB"
02 CLRG
03 CF 22
04 "UNITS?"
05 AON
06 PROMPT
07 AOFF
08 ASTO X
09 GTO IND X
```

Store units

```
10*LBL "SI"
11 5954.4
12 STO 01
13 14388
14 STO 02
15 2897.8
16 STO 03
17 5.6693 E-12
18 STO 04
19 GTO 00
20*LBL "EN"
21 18887982
22 STO 01
23 25998.4
24 STO 02
25 5216
26 STO 03
27 .171312 E-08
28 STO 04
```

Input prompting

```
29*LBL 00
30 "TEMP?"
31 PROMPT
32 STO 05
33 "WAVELENGTH?"
34 PROMPT
35 STO 06
36 "SOLVE"
37 PROMPT
```

Calculate λmax

```
38*LBL A
39 RCL 03
40 RCL 05
41 /
42 "WL MAX="
43 ARCL X
44 PROMPT
45*LBL B
46 RCL 03
47 RCL 06
48 /
```

Calculate T(λmax)

```
49 "TEMP="
50 ARCL X
51 PROMPT
```

Calculate E_b total

```
52*LBL C
53 RCL 05
54 X↑2
55 X↑2
56 RCL 04
57 *
58 "EbTOT="
59 ARCL X
60 PROMPT
```

Calculate $E_{b\lambda}$

```
61*LBL D
62 RCL 01
63 ENTER↑
64 +
65 PI
66 *
67 RCL 06
68 5
69 Y↑X
70 /
71 RCL 02
72 RCL 06
73 /
74 RCL 05
75 /
76 E↑X
77 1
78 -
79 /
80 "EbL="
81 ARCL X
82 PROMPT
```

Calculate $E_b(0-\lambda)$

```
83*LBL E
84 0
85 STO 08
86 STO 07
87*LBL 01
88 RDN
89 CLX
90 RCL 08
91 RCL 02
92 RCL 05
93 /
94 -
95 STO 08
96 3
97 X<>Y
```

Program Listings

```
 98 /
 99 RCL 06
100 X↑2
101 /
102 LASTX
103 1/X
104 RCL 06
105 /
106 -
107 6
108 RCL 06
109 /
110 RCL 08
111 X↑2
112 /
113 -
114 6
115 RCL 08
116 X↑2
117 /
118 RCL 08
119 /
120 +
121 RCL 08
122 RCL 06
123 /
124 E↑X
125 *
126 RCL 08
127 /
128 ST+ 07
129 RCL 07
130 /
131 1 E-05
132 X<=Y?
133 GTO 01
134 RDN
135 CLX
136 RCL 07
137 ENTER↑
138 +
139 PI
140 *
141 RCL 01
142 *
143 FS?C 00
144 RTN
145 "Eb0-L="
146 ARCL X
147 PROMPT
148♦LBL F
149 "WV LNTH
 2?"
150 PROMPT
151 ENTER↑
152 ENTER↑
153 SF 00
154 XEQ E
155 X<>Y
156 RCL 06
157 STO 00
158 RDN
159 STO 06
160 SF 00
161 XEQ E
162 -
163 ABS
164 RCL 00
165 STO 06
166 RDN
167 "EbL-L="
168 ARCL X
169 PROMPT
170 .END.
```

Calculate $E_b(\lambda_1-\lambda_2)$

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	λ	50	
	C_1		
	C_2		
	C_3		
	σ		
05	T	55	
	λ,λ'		
	sum		
	kc_2/T		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 57 USER MODE
ENG ____ FIX 4 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Used	
22		Used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BLACK HOLE CHARACTERISTICS

This program computes the Schwarzchild radius, lifetime and temperature of a black hole.

A black hole of mass (M) in grams has a Schwarzchild radius (R_s) in centimeters of:

$$R_s = \frac{2GM}{c^2} = M \bullet 1.484986855 \times 10^{-28}$$

where G is the universal gravitational constant and c is the speed of light.

The lifetime of a black hole (t_L) in seconds is given by:

$$t_L = M \bullet (10^{-28})$$

The temperatures of a black hole (K) in degrees Kelvin is:

$$K = \frac{10^{26}}{M}$$

Notes:

M must be greater than zero.

Underflow occurs for R_s when $M < 6.734066343 \times 10^{-72}$
t_L $M < 2.154434653 \times 10^{-24}$

Overflow occurs for K when $M < 1.000000001 \times 10^{-74}$
t_L $M > 2.154434650 \times 10^{33}$
M $R_s > 1.484986854 \times 10^{72}$

Example:

What is the temperature, Schwarzchild radius, and lifetime of a black hole with a mass of 1.99×10^{33} gm?

Keystrokes:	Display:
[USER]	(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] [ALPHA] HOLE [ALPHA]	ENTER KNOWN
1.99 [EEX] 33 [D]	MASS=1.9900E33
[B]	TEMP=5.0251E-8
[A]	RAD=295,512.3832
[C]	LIFE=7.8806E71

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program and set USER mode.		[USER]	
2.	Initialize the program.		[XEQ] HOLE	ENTER KNOWN
3.	Enter any one of the following:			
	Schwartzchild radius	R_s	[A]	RAD=
	Temperature	K	[B]	TEMP=
	Lifetime	t_L	[C]	LIFE=
	Mass	M	[D]	MASS=
4.	Compute any or all of the following:			
	Schwarzchild radius		[A]	RAD=
	Temperature		[B]	TEMP=
	Lifetime		[C]	LIFE=
	Mass		[D]	MASS=
	For new case, go to step 2.			

Program Listings

```
 01♦LBL "HOL
E"
 02 CF 22
 03 "ENTER K
NOWN"
 04 AVIEW
 05 RTN
```

Store and calculate mass

```
 06♦LBL D
 07 FS?C 22
 08 GTO 00
 09 RCL 03
 10♦LBL 04
 11 "MASS="
 12 ARCL X
 13 AVIEW
 14 RTN
 15♦LBL 00
 16 STO 03
 17 GTO 04
```

Store and calculate radius

```
 18♦LBL A
 19 FS?C 22
 20 GTO 01
 21 1.484986
85 E-28
 22 RCL 03
 23 *
 24 STO 00
 25♦LBL 05
 26 "RAD="
 27 ARCL X
 28 AVIEW
 29 RTN
 30♦LBL 01
 31 STO 00
 32 6.734066
34 E27
 33 *
 34 STO 03
 35 RCL 00
 36 GTO 05
```

Store and calculate temperature

```
 37♦LBL B
 38 FS?C 22
 39 GTO 02
 40 1 E26
 41 RCL 03
 42 /
 43 STO 01
 44♦LBL 06
 45 "TEMP="
 46 ARCL X
 47 AVIEW
 48 RTN
 49♦LBL 02
 50 STO 01
 51 1 E26
 52 X<>Y
 53 /
 54 STO 03
 55 RCL 01
 56 GTO 06
```

Store and calculate Lifetime

```
 57♦LBL C
 58 FS?C 22
 59 GTO 03
 60 1 E-28
 61 RCL 03
 62 3
 63 Y↑X
 64 *
 65 STO 02
 66♦LBL 07
 67 "LIFE="
 68 ARCL X
 69 AVIEW
 70 RTN
 71♦LBL 03
 72 STO 02
 73 1 E28
 74 *
 75 3
 76 1/X
 77 Y↑X
 78 STO 03
 79 RCL 02
 80 GTO 07
 81 .END.
```

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Radius	50	
	Temperature		
	Lifetime		
	Mass		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 29 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SPECIAL RELATIVITY CONVERSIONS

This program provides relativistic conversions between the following quantities: rest mass (m, in MeV); velocity (v, in units of c=1); energy (E, in MeV), and momentum (P). Given any two, it is possible to find the two unknowns by the following equations:

$$E = m/SQRT(1-v^2)$$

$$E = SQRT\ (P^2 + m^2)$$

$$E = P/v$$

Data may be entered in any order and recalled at any time. The program scans the registers and, after determining if there is enough data to solve for the unknown, selects the appropriate subset of equations.

Notes:

All data must be positive. Velocity must be less than 1. "DATA ERROR" message will be displayed if a real solution does not exist or the input data is outside these limits.

Examples:

1) Find the velocity and momentum of an electron (rest mass = .511 MeV) with a total energy of 1.0 MeV.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] SRC [ALPHA]	ENTER KNOWNS
.511 [B]	.5110
1 [D]	1.0000
[A]	V=0.8596
[C]	P=0.8596

2) At .9c, an electron has an energy of 1.1723 MeV. Find its rest mass and momentum.

Keystrokes:	Display:
[XEQ] [ALPHA] SRC [ALPHA]	ENTER KNOWNS
.9 [A]	0.9000
1.1723 [D]	1.1723
[B]	M=0.5110
[C]	P=1.0551

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize the program.		[XEQ] SRC	ENTER KNOWNS
3.	Enter the two values which are known:			
	velocity		[A]	(V)
	rest mass		[B]	(M)
	momentum		[C]	(P)
	energy		[D]	(E)
4.	Solve for the two unknowns:			
	velocity		[A]	V=
	rest mass		[B]	M=
	momentum		[C]	P=
	energy		[D]	E=

Program Listings

Initialize and prompt for knowns

```
01♦LBL "SRC
"
02 CLRG
03 SF 27
04 CF 22
05 CF 00
06 CF 01
07 CF 02
08 CF 03
09 CF 04
10 CF 05
11 CF 06
12 "ENTER K
NOWNS"
13 PROMPT
```

Storage and calculation of velocity

```
14♦LBL A
15 1
16 XEQ 00
17 RCL 02
18 X=0?
19 GTO 01
20 RCL 03
21 X≠0?
22 GTO 15
23 XEQ 06
24 ENTER↑
25♦LBL 07
26 RDN
27 RCL 04
28 X<>Y
29 /
30 SF 00
31 GTO c
```

Storage and calculation of rest mass

```
32♦LBL B
33 2
34 XEQ 00
35 X=0?
36 GTO 04
37 RCL 03
38 *
39 LASTX
40 X≠0?
41 GTO 02
42 RCL 04
43 ENTER↑
44 ENTER↑
45 RCL 01
46 /
47♦LBL 02
48 ENTER↑
49 RDN
50 XEQ 03
51 R↑
52 *
53 SF 00
54 GTO c
```

SQRT $(1-y^2/x^2)$

```
55♦LBL 03
56 /
57 ASIN
58 COS
59 FS?C 05
60 GTO 10
61 RTN
```

Storage and calculation of energy

```
62♦LBL D
63 3
64 XEQ 00
65 X=0?
66 GTO 06
67 RCL 02
68 X=0?
69 GTO 07
70 FC? 03
71 SF 05
72 SF 06
73♦LBL 08
74 X<>Y
75 ASIN
76 COS
77 /
78 FS?C 05
79 GTO 11
80 FS?C 06
81 GTO 12
82 RTN
```

Storage and calculation of momentum

```
83♦LBL C
84 4
85 XEQ 00
86 X=0?
87 GTO 09
88 RCL 02
89 X≠0?
90 XEQ 08
91 RCL 03
92 X=0?
93 RDN
94 RCL 01
95 *
96 SF 00
```

Check for valid answer

```
97♦LBL c
98 X=0?
99 /
100 FC?C 00
```

Program Listings

```
101 RTN
```

Select output routine

```
102 FS?C 01
103 GTO 10
104 FS?C 02
105 GTO 11
106 FS?C 03
107 GTO 12
108 FS?C 04
109 GTO 13
```

Velocity output

```
110♦LBL 10
111 CF 01
112 "V="
113 ARCL X
114 PROMPT
```

Rest mass output

```
115♦LBL 11
116 CF 02
117 "M="
118 ARCL X
119 PROMPT
```

Energy output

```
120♦LBL 12
121 CF 03
122 "E="
123 ARCL X
124 PROMPT
```

Momentum output

```
125♦LBL 13
126 CF 04
127 "P="
128 ARCL X
129 PROMPT
130♦LBL 00
131 STO 00
132 FS?C 22
133 GTO b
134 SF IND 0
0
135 RCL IND
00
136 X≠0?
137 STOP
138 RCL 01
139 RTN
```

Indirect storage routine

```
140♦LBL b
141 RDN
142 STO IND
00
143 STOP
144♦LBL 01
145 RCL 03
146 ENTER↑
147 GTO 07
148♦LBL 04
149 RCL 04
150 GTO d
151♦LBL 06
152 RCL 02
153 XEQ c
154 RCL 04
155 XEQ c
156 R-P
157 RTN
158♦LBL 09
159 RCL 02
160♦LBL d
161 XEQ c
162 RCL 03
163 GTO 02
164♦LBL 15
165 SF 05
166 GTO 03
167 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Indirect	50	
	Velocity		
	Rest Mass		
	Total Energy		
	Momentum		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 45 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

THREE DIMENSIONAL SPECIAL RELATIVITY

(requires one memory module)

Given an event as the components of a 4-vector $x'^{\mu} = (x', y', z', ct')$ in a frame of reference moving at a fraction of the speed of light $\beta = (\beta^x, \beta^y, \beta^z) = \overline{V}/c$, this program will calculate the speed, $|\beta|$, the time dilation factor, γ, the 4-vector of the event in the fixed frame, x^{μ}, and the invariant interval, $c\Delta\tau$.

Formulae:

$$\Delta\bar{x} = \Delta\bar{x}' + \bar{\beta}[(\gamma-1)\frac{\bar{\beta}\ \bar{x}'}{\beta^2} + \gamma c\Delta t']$$

$$\Delta t = \gamma(t + \bar{\beta}\ \bar{x}/c^2)$$

$$\gamma = [1-\beta^2]^{-\frac{1}{2}}$$

$$\beta = |\bar{\beta}| = [(\beta^x)^2+(\beta^y)^2+(\beta^z)^2]^{\frac{1}{2}}$$

$$|\Delta\bar{x}| = [\Delta x^2+\Delta y^2+\Delta z^2]^{\frac{1}{2}}$$

$$c^2\Delta\tau^2 = c^2\Delta t^2-|\Delta\bar{x}|^2$$

Notes:

The coordinate frames are assumed to be synchronized so that the event (0,0,0,0,) has the same coordinates in both frames.

For a spacelike interval, $c^2\Delta\tau^2<0$, the calculator will display " - $|c\Delta\tau|$"

Example:

An observer moving relative to earth with velocity $\vec{\beta} = (.4, .5, .6)$ measures the coordinates of an event as $x'^{\mu} = (1, 2, 3, 4)$.

a) What are the coordinates of the event relative to the earth frame?

b) What is the interval between the event and the origin (0,0,0,0,)?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 016	
[XEQ] [ALPHA] 3D [ALPHA]	DIMENSIONS?
3 [R/S]	BX?
.4 [R/S]	BY?
.5 [R/S]	BZ?
.6 [R/S]	SPEED=0.8775
[R/S]	D.F.=2.0851
[R/S]	X PRIME?
1 [R/S]	Y PRIME?
2 [R/S]	Z PRIME?
3 [R/S]	cT PRIME?
4 [R/S]	cT=15.0130
[R/S]	Z=10.7102
[R/S]	Y=8.4251
[R/S]	X=6.1401
[R/S]	cdT=1.4142

User Instructions

SIZE: 016

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize the program.		[XEQ] 3D	DIMENSIONS?
3.	Input the number of dimensions	n	[R/S]	BX?
4.	Input velocity in x direction.	β^x	[R/S]	BY?
5.	Input velocity in the Y direction.	β^y	[R/S]	BZ?
6.	Input velocity in the Z direction and			
	compute speed.	β^z	[R/S]	SPEED=
7.	Compute the corresponding dilation			
	factor.		[R/S]	D.F.=
8.	Continue the program execution.		[R/S]	X PRIME?
9.	Input x' coordinate of event.	x'	[R/S]	Y PRIME?
10.	Input y' coordinate of event.	y'	[R/S]	Z PRIME?
11.	Input z' coordinate of event.	z'	[R/S]	cT PRIME?
12.	Input cT' coordinate of event and calculate			
	relative cT coordinate of event.	cT'	[R/S]	cT=
13.	Compute relative Z coordinate.		[R/S]	Z=
14.	Compute relative Y coordinate.		[R/S]	Y=
15.	Compute relative X coordinate.		[R/S]	X=
16.	Compute $c\Delta\tau$.		[R/S]	cdT=
	For new case go to step 2.			

Program Listings

```
01♦LBL "3D"
02 "DIMENSI
ONS?"
03 PROMPT
04 XEQ E
05 CF 03
06 SF 02
07 FS? 00
08 CF 02
09 FS? 01
10 CF 02
11 FS? 01
12 SF 03
13 FS? 02
14 SF 03
```

Input prompting for $\bar{\beta}$

```
15 "BX?"
16 FS? 02
17 PROMPT
18 "BY?"
19 FS? 03
20 PROMPT
21 "BZ?"
22 PROMPT
23 XEQ A
```

Input prompting for x'^{μ}

```
24 CLST
25 "X PRIME
?"
26 FS? 02
27 PROMPT
28 "Y PRIME
?"
29 FS? 03
30 PROMPT
31 "Z PRIME
?"
32 PROMPT
33 "cT PRIM
E?"
34 PROMPT
```

Calculate x^{μ}

```
35 XEQ B
36 XEQ e
37 RCL 13
38 X↑2
39 /
40 RCL 14
41 1
42 -
43 *
44 RCL 14
45 RCL 07
46 *
47 +
48 STO 15
49 3
50 STO 00
51 STO 12
52♦LBL 00
53 RCL 00
54 STO 12
55 RCL 15
56 RCL IND
12
57 *
58 RCL 12
59 3
60 +
61 STO 12
62 CLX
63 RCL IND
12
64 +
65 5
66 ST+ 12
67 CLX
68 RDN
69 STO IND
12
70 5
71 ST- 12
72 2
73 FS? 00
74 ST- 00
75 .5
76 FS? 01
77 ST- 00
78 RCL 00
79 1
80 -
81 STO 00
82 X≠0?
83 GTO 00
84 RCL 07
85 XEQ e
86 +
87 RCL 14
88 *
89 STO 08
90 "cT="
91 ARCL X
92 PROMPT
93 "Z="
94 ARCL 11
```

Program Listings

```
 95 PROMPT
 96 FS? 00
 97 GTO D
 98 "Y="
 99 ARCL 09
100 PROMPT
101 FS? 01
102 GTO D
103 "X="
104 ARCL 10
105 PROMPT
106 GTO D
```

Calculate $|\bar{\beta}|$ and γ .

```
107•LBL A
108 STO 03
109 FS? 00
110 GTO 06
111 X<>Y
112 R-P
113 LASTX
114 STO 01
115 X<>Y
116 FS? 01
117 GTO 06
118 R↑
119 STO 02
120 R-P
121•LBL 06
122 STO 13
123 ACOS
124 SIN
125 1/X
126 STO 14
127 RCL 13
128 "SPEED="
129 ARCL X
130 PROMPT
131 "D.F.="
132 ARCL Y
133 PROMPT
134 RTN
```

Store x'^{μ}

```
135•LBL B
136 STO 07
137 RDN
138 STO 06
139 RDN
140 STO 04
141 RDN
142 STO 05
143 RTN
```

Set flags for correct no. of dimensions

```
144•LBL E
145 STO 12
146 GTO IND
12
147•LBL 01
148 CF 01
149 SF 00
150 RTN
151•LBL 02
152 CF 00
153 SF 01
154 RTN
155•LBL 03
156 CF 00
157 CF 01
158 RTN
```

Calculate $\Sigma\beta^{i}x'^{i}$

```
159•LBL e
160 RCL 06
161 RCL 03
162 *
163 FS? 00
164 RTN
165 RCL 01
166 RCL 04
167 *
168 +
169 FS? 01
170 RTN
171 RCL 02
172 RCL 05
173 *
174 +
175 RTN
```

Compute $c\Delta\tau$

```
176•LBL D
177 RCL 06
178 FS?C 00
179 GTO 04
180 RCL 04
181 R-P
182 FS?C 01
183 GTO 04
184 RCL 05
185 R-P
186•LBL 04
187 X↑2
188 RCL 07
189 X↑2
190 -
191 CHS
192 CF 04
193 X<0?
194 SF 04
195 ABS
```

```
SQRT
FS?C 04
CHS
"cdT="
ARCL X
AVIEW
RTN
.END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	counter	50	
	VEL y		
	VEL x		
	VEL z		
	y		
05	x	55	
	z		
	ct		
	T		
	Y'		
10	X'	60	
	Z'		
	indirect		
	|β|		
	γ		
15	used	65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 016 TOT. REG. 67 USER MODE
ENG ___ FIX 4 SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		n=1	
01		n=2	
02		n>2	
03		n>1	
04		used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EINSTEINS TWIN PARADOX

The program is arranged to calculate subjective and real time differential between an observer on Earth and the pilot of a vehicle accelerating near the speed of light. Imagine the twins at age 21. One becomes an astronaut and volunteers for the first interstellar flight. He takes off and travels at a speed of 2.99×10^8 meters per second. The astronaut travels for what he measures to be a year, at which time he fires retro and navigational engines and turns around and heads toward Earth. The journey naturally takes another year. He is now 23 years old; but when he steps from the ship, his twin is over 48 years old!

Equation:

$$T_S = T_E \sqrt{1 - \frac{v^2}{c^2}}$$

where:

T_S=time passed on board ship

T_E=time passed on earth

v =ship velocity relative to earth

c =speed of light (2.9979×10^8 m/s)

Notes:

Be certain that the speed of the spacecraft is in meters per second.

Example:

If one twin travels for 10 years at 87% the speed of light, how much time will have passed for the twin left on earth?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 003	
[XEQ] [ALPHA] TWIN [ALPHA]	READY
2.9979 [EEX] 8 [ENTER↑]	
.87 [X] [A]	V=260,817,300.0
10 [B]	TS=10.0000
[C]	TE=20.2818

If the ship had only been going 1 X 10^4 m/s, how much time would have passed for the earth-bound twin?

[EEX] 4 [A]	V=10,000.0000
[C]	TE=10.0000

User Instructions

SIZE: 003

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize.		[XEQ]TWIN	READY
3	Key in any two variables:			
	ship velocity	v (m/s)	[A]	V=
	ship time	T_S	[B]	TS=
	Earth time	T_E	[C]	TE=
4	Solve for the unknown:			
	ship velocity		[A]	V=
	ship time		[B]	TS=
	Earth time		[C]	TE=

Program Listings

01♦LBL "TWIN" 02 CF 22 03 SF 27 04 "READY" 05 AVIEW 06 RTN	Initialization
07♦LBL A 08 FS?C 22 09 GTO 00 10 1 11 RCL 01 12 RCL 02 13 / 14 X↑2 15 - 16 SQRT 17 2.9979 E8 18 * 19♦LBL 00 20 STO 00 21 "V=" 22 GTO 02	Velocity
23♦LBL B 24 FS?C 22 25 GTO 00 26 XEQ 01 27 RCL 02 28 * 29♦LBL 00 30 STO 01 31 "TS=" 32 GTO 02	Ship time
33♦LBL C 34 FS?C 22 35 GTO 00 36 XEQ 01 37 1/X 38 RCL 01 39 * 40♦LBL 00 41 STO 02 42 "TE="	Earth time
43♦LBL 02 44 ARCL X 45 AVIEW 46 RTN	General output
47♦LBL 01 48 1 49 RCL 00 50 2.9979 E8 51 / 52 X↑2 53 - 54 SQRT 55 RTN 56 .END.	Lorentz transform

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	V	50	
	TS		
	TE		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 003 TOT. REG. 18 USER MODE

ENG ____ FIX 4 SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

DELTA-V ORBIT SIMULATOR

(Requires One Memory Module)

This program calculates orbit parameters from initial position and velocity data both for elliptical and hyperbolic orbits in a plane. It is also possible to move the point of interest to anywhere along the orbit and then recalculate orbit parameters.

Equations Used:

$$E = \frac{1}{2} V_i^{\,2} - \frac{Gm}{R_i}$$

$$\ell = VR \sin(\alpha_i - \theta_i)$$

$$\varepsilon = \sqrt{1 - \frac{2E\ell^2}{(Gm)^2}}$$

$$R_o = \frac{\ell^2}{Gm}$$

$$\theta' = \theta_i + \cos^{-1}\left(\frac{\frac{R_o}{R_i} - 1}{\varepsilon}\right)$$

$$R_{min} = \frac{R_o}{1 + \varepsilon}$$

$$a = \frac{R_o}{1 - \varepsilon^2}$$

$$b = a\sqrt{1 - \varepsilon^2}$$

$$T = 2\pi\sqrt{\frac{a^3}{Gm}}$$

$$S = \frac{R_{min}}{1 - \frac{1}{\varepsilon}}$$

$$\theta_a = \cos^{-1}\left(\frac{1}{\varepsilon}\right)$$

Given θ_{new} :

$$R_{new} = \frac{R_o}{1 + \varepsilon \cos(\theta_{new} - \theta')}$$

$$V_{new} = \sqrt{2\left(E + \frac{Gm}{R_{new}}\right)}$$

$$\alpha_{new} = \theta_{new} + \sin^{-1}\left(\frac{\ell}{V_{new} R_{new}}\right)$$

For change in velocity:

$$\overline{V_{new}} = \overline{V_{old}} + \overline{\Delta V}$$



ε = 0   ε < 1   (e > 1)

Where:

E = Energy

ℓ = Angular momentum

ε = Eccentricity

a = Semimajor axis

b = Semiminor axis

T = Period

S = Distance to asymptote vertex

θ_a = Angle between asymptotes and radius vector

α = Velocity angle

Example:

Execute a transfer from low earth orbit to high earth orbit.

$R_{init} = 7.1E6m$, $\theta_{init} = 0°$, $V_{init} = 7.4E3$ m/s, V∡ = 90°

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] ORBIT [ALPHA]	MASS ?
5.979 [EEX] 24 [R/S]	INIT. R ?
7.1 [EEX] 6 [R/S]	INIT. ∡ ?
0 [R/S]	INIT. V ?
7.4 [EEX] 3 [R/S]	INIT. V∡ ?
90 [R/S]	R=7100000.00
90 [ENTER↑]	
2300 [D]	R=7100000.000
[B]	E=0.6743
[R/S]	∡P=359.9976
359.9976 [ENTER↑]	
180 [+] [C]	R=36501420.27
[R/S]	∡=179.9976
[R/S]	V=1886.7759
[R/S]	V∡=270.0009
270.0009 [ENTER↑]	
1450 [D]	R=36501420.27
[R/S]	∡=179.9976
[R/S]	V=3336.7759
[R/S]	V∡=270.0009
[B]	E=0.0186
[R/S]	∡P=180.1643
[R/S]	RMIN=36501417.46
[R/S]	ELLIPSE
[R/S]	A=37192989.78
[R/S]	B=37186559.62
[R/S]	T=19.8193

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize.		[XEQ] ORBIT	MASS ?
3	Key in the mass of the attracting body.	M	[R/S]	INIT. R ?
4	Key in the initial distance from center of			
	attracting body.	R_{init}	[R/S]	INIT. ∡ ?
5	Key in the initial longitude.	θ_{init}	[R/S]	INIT. V ?
6	Key in the initial velocity.	V_{init}	[R/S]	INIT. V∡ ?
7	Key in the initial angle of velocity.	$\theta_{v_{init}}$	[R/S]	
8	Output current position.			R=
			[R/S]	∡= (θ)
			[R/S]	V=
			[R/S]	V∡= (α)
9	To output orbit geometry:		[B]	RMIN=
				∡P=(θ')
a	If circular:			CIRCLE
				R=
				T= (hrs)
b	If elliptical:			ELLIPSE
				A=
				B=
				T= (hrs)
c	If parabolic:			PARABOLA
d	If hyperbolic:			HYPERBOLA
				S=
				∡A=(θa)
10	To move to a new position in the old orbit,	θ_{new}	[C]	R=
	Go to step 8.			

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
11	To change the velocity vector:			
	Key in angle of change.	θ	[ENTER↑]	
	Key in magnitude of change.	ΔV(^{m}/S)	[D]	R=
	Go to step 8.			
12	For new initial conditions:		[A]	MASS ?
	go to step 3.			

Program Listings

01♦LBL "ORBIT"	Initialize	50 RCL 09	
02 SF 21		51 *	
03 SF 27		52 STO 05	
04♦LBL A		53 "RMIN="	
05 CF 22	Input	54 ARCL 05	R_{min}
06 "MASS ?"		55 AVIEW	
07 PROMPT		56 RCL 08	
08 6.6732 E-11		57 ENTER↑	
09 *		58 FRC	
10 FS?C 22		59 X≠Y?	
11 STO 04		60 GTO 01	e≥1
12 "INIT. "		61 RCL 09	
13 ASTO 05		62 1	
14 "⊢R ?"		63 RCL 08	
15 PROMPT		64 X=0?	
16 FS?C 22		65 GTO 03	e=0
17 STO 00		66 "ELLIPSE"	Ellipse
18 CLA		67 AVIEW	
19 ARCL 05		68 X↑2	
20 "⊢∡ ?"		69 -	
21 PROMPT		70 /	
22 FS?C 22		71 ENTER↑	
23 STO 01		72 "A="	
24 CLA		73 ARCL X	
25 ARCL 05		74 AVIEW	
26 "⊢V ?"		75 LASTX	
27 PROMPT		76 SQRT	
28 FS?C 22		77 *	
29 STO 02		78 "B="	
30 CLA		79 ARCL X	
31 ARCL 05		80 AVIEW	
32 "⊢V∡ ?"		81 RDN	
33 PROMPT		82 GTO 00	
34 FS?C 22		83♦LBL 03	Circle
35 STO 03		84 "CIRCLE"	
36 XEQ 51		85 AVIEW	
37 GTO 09		86 "R="	
38♦LBL B		87 ARCL 05	
39 "E="	Review geometry	88 RCL 05	
40 ARCL 08	of orbit	89 AVIEW	
41 AVIEW		90♦LBL 00	Period calculation
42 "∡P="	θ'	91 3	
43 ARCL 10		92 Y↑X	
44 RCL 10		93 RCL 04	
45 AVIEW		94 /	
46 RCL 08		95 SQRT	
47 1		96 2	
48 +		97 *	
49 1/X		98 PI	
		99 *	

Program Listings

```
100 3600
101 /
102 "T="
103 ARCL X
104 AVIEW
105 STOP
106 GTO B
```

Hyperbola or parabola

```
107♦LBL 01
108 RCL 05
109 1
110 RCL 08
111 1/X
112 -
113 X≠0?
114 GTO 00
115 "PARABOL
A"
116 AVIEW
117 STOP
118 GTO B
119♦LBL 00
120 "HYPERBO
LA"
121 AVIEW
122 /
123 "S="
124 ARCL X
125 AVIEW
126 RCL 08
127 1/X
128 ACOS
129 "∡A="
130 ARCL X
131 AVIEW
132 STOP
133 GTO B
```

Move to new θ

```
134♦LBL C
135 XEQ 50
136 STO 01
137 RCL 10
138 -
139 COS
140 RCL 08
141 *
142 1
143 +
144 RCL 09
145 /
146 1/X
147 STO 00
```

R_{new}

```
148 RCL 04
149 X<>Y
150 /
151 RCL 06
152 +
153 2
154 *
155 SQRT
156 STO 02
```

V_{new}

```
157 RCL 07
158 RCL 00
159 RCL 02
160 *
161 /
162 INT
163 X=0?
164 LASTX
165 ASIN
166 RCL 07
167 RCL 01
168 RCL 10
169 -
170 SIN
171 *
172 X>0?
173 GTO 00
174 CLX
175 180
176 -
177 CHS
178 R↑
179♦LBL 00
180 RDN
181 RCL 01
182 +
183 XEQ 50
184 STO 03
185 GTO 09
```

α_{new}

Change velocity

```
186♦LBL D
187 P-R
188 RCL 03
189 RCL 02
190 P-R
191 ST+ Z
192 RDN
193 ST+ Z
194 RDN
195 R-P
196 STO 02
197 RDN
198 XEQ 50
199 STO 03
```

Program Listings

Program	Comments
200 XEQ 51	
201◆LBL 09	Output position and velocity
202 "R="	
203 ARCL 00	
204 AVIEW	
205 "∡="	
206 ARCL 01	
207 AVIEW	
208 "V="	
209 ARCL 02	
210 AVIEW	
211 "V∡="	
212 ARCL 03	
213 AVIEW	
214 STOP	
215 GTO 09	
216◆LBL 50	Scale an angle
217 1	
218 P-R	
219 R-P	
220 RDN	
221 X≠0?	
222 X>0?	
223 RTN	
224 360	
225 +	
226 RTN	
227◆LBL 51	Calculate new orbit geometry
228 RCL 02	
229 X↑2	
230 2	
231 /	
232 RCL 04	
233 RCL 00	
234 /	
235 -	
236 STO 06	Energy
237 RCL 03	
238 RCL 01	
239 -	
240 SIN	
241 RCL 00	
242 *	
243 RCL 02	
244 *	
245 STO 07	Angular momentum
246 X↑2	
247 2	
248 *	
249 RCL 06	
250 *	
251 RCL 04	
252 X↑2	
253 /	
254 1	
255 +	
256 SQRT	
257 STO 08	Eccentricity
258 RCL 01	
259 RCL 07	
260 X↑2	
261 RCL 04	
262 /	
263 STO 09	R_o
264 RCL 00	
265 /	
266 1	
267 -	
268 RCL 08	
269 /	
270 INT	
271 X≠0?	
272 GTO 00	
273 CLX	
274 LASTX	
275◆LBL 00	
276 ACOS	
277 RCL 03	
278 RCL 01	
279 -	
280 COS	
281 LASTX	
282 SIN	
283 *	
284 X<0?	
285 GTO 00	
286 RDN	
287 CHS	
288 R↑	
289◆LBL 00	
290 RDN	
291 +	
292 XEQ 50	
293 STO 10	θ'(ΔP)
294 .END.	

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Dist (R)	50	
	Bearing (∠)		
	Speed (V)		
	Speed bearing (V∠)		
	G*M		
05	Temp	55	
	Energy		
	Ang. mom.		
	e		
	R_o		
10	θ' (∠P)	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 77 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EQUATIONS OF MOTION

This program provides an interchangeable solution between time, displacement, final velocity, initial velocity, and acceleration. Given any three knowns, the two unknowns will be calculated. The motion must be linear and have constant acceleration.

Equations:

$$x = \frac{t(VF+VI)}{2}$$

$$x = VF(t) - \frac{1}{2}\, a\, t^2$$

$$x = \frac{VF^2 - VI^2}{2a}$$

$$x = VI(t) + \frac{1}{2}\, a\, t^2$$

$$VF = VI + a\, t$$

where:

t = time

x = displacement

VF = final velocity

VI = initial velocity

a = acceleration

Notes:

Any consistent set of units may be used.
Displacement, acceleration, and velocity should be considered as signed quantities. For example: if VI and a are in opposite directions, one should be input as negative.
All equations assume initial displacement, X_o, equals zero.

Example:

If a rock is dropped off a 50 foot cliff, how long does it take to drop, and how fast is it going when it hits bottom? ($a=32.16$ ft/s^2)

Keystrokes:	Display
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] MOTION [ALPHA]	T?
[R/S]	X?
50 [R/S]	VF?
[R/S]	VI?
0 [R/S]	A?
32.16 [R/S]	T=1.7634
[R/S]	X=50.0000
[R/S]	VF=56.7098
[R/S]	VI=0.0000
[R/S]	a=32.1600

(time = 1.7634 seconds and final velocity = 56.7098 feet/second)

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] MOTION	T?
3.	Key in time if known.		[R/S]	X?
4.	Key in displacement if known.		[R/S]	VF?
5.	Key in final velocity if known.		[R/S]	VI?
6.	Key in initial velocity if known.		[R/S]	A?
7.	Key in acceleration if known.		[R/S]	T=
	(3 of the inputs above must be made. The			
	2 unknowns are specified by simply press-			
	ing [R/S].)			
8.	Continue with the output of all quantities.		[R/S]	X=
			[R/S]	VF=
			[R/S]	VI=
			[R/S]	A=

Program Listings

01♦LBL "MOTION"	Initialize
02 CF 00	
03 CF 01	
04 CF 02	
05 CF 03	
06 CF 04	
07 CF 22	
08 "T?"	
09 PROMPT	Input
10 FS?C 22	
11 SF 00	
12 STO 00	
13 "X?"	
14 PROMPT	
15 FS?C 22	
16 SF 01	
17 STO 01	
18 "VF?"	
19 PROMPT	
20 FS?C 22	
21 SF 02	
22 STO 02	
23 "VI?"	
24 PROMPT	
25 FS?C 22	
26 SF 04	
27 STO 04	
28 "A?"	
29 PROMPT	
30 FS?C 22	
31 SF 03	
32 STO 03	
33 FS? 01	x known?
34 GTO 01	
35 FS? 02	
36 GTO 00	VF known?
37 RCL 04	
38 RCL 00	t, a and VI known
39 *	
40 RCL 03	
41 RCL 00	
42 X↑2	
43 *	
44 2	
45 /	
46 +	
47 STO 01	x
48 SF 01	
49 GTO 02	
50♦LBL 00	a known?
51 FS? 03	
52 GTO 00	
53 RCL 04	
54 RCL 02	t, VF and VI known
55 +	
56 2	
57 /	
58 RCL 00	
59 *	
60 STO 01	x
61 SF 01	
62 GTO 09	
63♦LBL 00	
64 FS? 00	t known?
65 GTO 00	
66 RCL 02	
67 X↑2	VF, a and VI known
68 RCL 04	
69 X↑2	
70 -	
71 RCL 03	
72 2	
73 *	
74 /	
75 STO 01	x
76 SF 01	
77 GTO 05	
78♦LBL 00	
79 RCL 02	t, VF and a known
80 RCL 00	
81 *	
82 RCL 03	
83 RCL 00	
84 X↑2	
85 *	
86 2	
87 /	
88 -	
89 STO 01	x
90 SF 01	
91 GTO 08	
92♦LBL 01	
93 FS? 02	VF known?
94 GTO 03	
95 FS? 03	
96 GTO 00	a known?
97 RCL 01	t, x and VI known
98 RCL 00	
99 /	
100 2	
101 *	

Program Listings

102 RCL 04	
103 -	
104 STO 02	VF
105 SF 02	
106 GTO 07	
107♦LBL 00	
108 FS? 00	t known
109 GTO 02	
110 RCL 01	
111 RCL 03	x, a and VI known
112 *	
113 2	
114 *	
115 RCL 04	
116 X↑2	
117 +	
118 SQRT	
119 RCL 01	
120 SIGN	
121 *	
122 STO 02	VF
123 SF 02	
124 GTO 05	
125♦LBL 02	x, a and t known
126 RCL 01	
127 RCL 00	
128 /	
129 RCL 03	
130 RCL 00	
131 *	
132 2	
133 /	
134 +	
135 STO 02	VF
136 SF 02	
137 GTO 08	
138♦LBL 03	
139 FC? 03	a unknown?
140 GTO 00	
141 RCL 02	
142 X↑2	VF, a and x known
143 RCL 01	
144 RCL 03	
145 *	
146 2	
147 *	
148 -	
149 SQRT	
150 RCL 01	
151 SIGN	
152 CHS	
153 *	
154 RCL 02	
155 +	
156 RCL 03	
157 /	
158 STO 00	t
159 SF 00	
160 GTO 08	
161♦LBL 00	
162 FS? 00	t known?
163 GTO 07	
164♦LBL 09	
165 RCL 02	VF, x and VI known
166 X↑2	
167 RCL 04	
168 X↑2	
169 -	
170 RCL 01	
171 /	
172 2	
173 /	
174 STO 03	a
175 SF 03	
176♦LBL 05	
177 RCL 01	VF, x and VI known
178 RCL 02	
179 RCL 04	
180 +	
181 /	
182 2	
183 *	
184 STO 00	t
185 SF 00	
186 GTO 10	
187♦LBL 07	
188 RCL 02	VF, x and t known
189 RCL 01	
190 RCL 00	
191 /	
192 -	
193 RCL 00	
194 /	
195 2	
196 *	
197 STO 03	a
198 SF 03	
199♦LBL 08	
200 RCL 02	VF, a and t known
201 RCL 03	
202 RCL 00	
203 *	

Program Listings

```
204 -
205 STO 04
206 SF 04
207♦LBL 10
208 "T="
209 ARCL 00
210 AVIEW
211 STOP
212 "X="
213 ARCL 01
214 AVIEW
215 STOP
216 "VF="
217 ARCL 02
218 AVIEW
219 STOP
220 "VI="
221 ARCL 04
222 AVIEW
223 STOP
224 "A="
225 ARCL 03
226 AVIEW
227 RTN
228 .END.
```

VI

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	t	50	
	x		
	VF		
	a		
	VI		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 51 USER MODE
ENG ____ FIX 4 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		t known	
01		x known	
02		VF known	
03		a known	
04		VI known	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ISOTOPE OVERLAP CORRECTIONS

This program corrects for spillover between channels when two radioactive isotopes are being counted in a liquid scintillation spectrometer. Background subtraction for each isotope is also provided. The program may be used with single isotope.

Isotopes x and y are counted in machine channels A and B, respectively.

Let a = fractioned spillover of isotope y from channel B to A.
b = fractioned spillover of isotope x from channel A to B.

C_x = corrected counts/min isotope x in channel A = $\dfrac{C_A - a\ C_B}{1-ab}$,

where C_A and C_B are the observed counts/min in each channel.

$$C_y = \frac{C_B - b\ C_A}{1-ab}$$

<u>Outputs</u>

Total counts/min isotope x = Cx (1+b) = Tx

Total counts/min isotope y = Cy (1+a) = Ty

Example:

2 isotopes, Spillover A → B = 10%, Spillover B → A = 20%, BKA = 10 CPM, BKB = 50 CPM.
For the following values of CPM A and CPM B, calculate the corrected values and totals.

Sample#	A	B
1	1,000	500
2	2,000	1,000
3	1,400	2,200

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 011		
[XEQ] [ALPHA] IOC [ALPHA]	ISOTOPES?	
2 [R/S]	SPLOVER B-A?	
.2 [R/S]	BKA?	
10 [R/S]	SPLOVER A-B?	
.1 [R/S]	BKB?	
50 [R/S]	CPM A?	
1,000 [R/S]	CPM B?	
500 [R/S]	TX=1,010.	
[R/S]	TY=430.	
[R/S]	NEXT=2	
[R/S]	CPM A?	
2,000 [R/S]	CPM B?	
1,000 [R/S]	TX=2,020	
[R/S]	TY=920	
[R/S]	NEXT=3	
[R/S]	CPM A?	
1,400 [R/S]	CPM B?	
2,200 [R/S]	TX=1,078	
[R/S]	TY=2,462	
[R/S]	NEXT=4	
[R/S]	CPM A?	
[R/S]	TOT. TX=4,108	(Pressing [R/S] with no input displays totals and resets for new case)
[R/S]	TOT. TY=3,812	

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize the program.		[XEQ] IOC	ISOTOPES?
3.	Input number of isotopes.	n	[R/S]	SPLOVER B-A?
4.	Input the spillover from B to A as a			
	fraction.	a	[R/S]	BKA?
5.	Input the background CPM on channel A.	BKA	[R/S]	SPLOVER A-B?
6.	Input the spillover from A to B as a			
	fraction.	b	[R/S]	BKB?
7.	Input the background CPM on channel B.	BKB	[R/S]	CPM A?
8.	Input first CPM for channel A.	CPM A	[R/S]	CPM B?
9.	Input first CPM for channel B.	CPM B	[R/S]	TX=
	Calculator will display Tx, Ty, the		[R/S]	TY=
	number of the next sample, and then prompt		[R/S]	NEXT=
	for the next CPMA.		[R/S]	CPM A?
10.	Go to step 8 until all samples have been		[R/S]	TOT. TX=
	entered, then press [R/S] to calculate		[R/S]	TOT. TY=
	totals and reset for new problem.		[R/S]	ISOTOPES?
	For single isotope, ignore steps 6, 7 and			
	9.			

Program Listings

Initialize and prompt for data input

```
01♦LBL "IOC"
02 CF 00
03 CLRG
04 2
05 "ISOTOPES?"
06 PROMPT
07 X=Y?
08 SF 00
09 2
10 STO 10
11 "SPLOVER B-A?"
12 PROMPT
13 ENTER↑
14 "BKA?"
15 PROMPT
16 XEQ D
17 FC? 00
18 GTO 00
19 "SPLOVER A-B?"
20 PROMPT
21 ENTER↑
22 "BKB?"
23 PROMPT
24 XEQ E
25♦LBL 00
26 CF 22
27 "CPM A?"
28 CLX
29 PROMPT
30 FC?C 22
31 GTO "TOT"
32 XEQ A
33 FC? 00
34 GTO 01
35 "CPM B?"
36 PROMPT
37 XEQ B
38♦LBL 01
39 GTO C
```

Subtract BKA from CPMA and store

```
40♦LBL A
41 RCL 00
42 -
43 STO 04
44 RTN
```

Subtract BKB from CPMB and store

```
45♦LBL B
46 RCL 02
47 -
48 STO 05
49 RTN
```

Calculate TX and TY

```
50♦LBL C
51 RCL 04
52 RCL 01
53 RCL 05
54 *
55 -
56 1
57 RCL 01
58 RCL 03
59 *
60 -
61 /
62 1
63 RCL 03
64 +
65 *
66 ST+ 06
67 STO 07
68 FIX 0
69 RCL 05
70 RCL 03
71 RCL 04
72 *
73 -
74 1
75 RCL 01
76 RCL 03
77 *
78 -
79 /
80 1
81 RCL 01
82 +
83 *
84 ST+ 08
85 STO 09
86 RCL 07
```

Display TX

```
87 "TX= "
88 ARCL X
89 AVIEW
90 STOP
91 RCL 09
92 FC? 00
93 GTO 00
```

Display TY

```
94 "TY= "
95 ARCL X
96 AVIEW
97 STOP
```

(cont'd.)

Program Listings

```
 98 RCL 10
 99 "NEXT= "
100 ARCL X
101 AVIEW
102 STOP
103 ADV
104 ISG 10
105 RCL 10
106 GTO 00
107♦LBL "TOT
"                          Display totals
108 RCL 06                 and return to
109 "TOT. TX               Label IOC
="
110 ARCL X
111 AVIEW
112 STOP
113 FC? 00
114 RTN
115 RCL 08
116 "TOT. TY
="
117 ARCL X
118 AVIEW
119 STOP
120 0
121 STO 10
122 GTO "IOC
"
123♦LBL D                  Store BKA
124 STO 00
125 RDN
126 STO 01
127 RTN
128♦LBL E                  Store BKB
129 STO 02
130 RDN
131 STO 03
132 RTN
133 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	BKA	50	
	a		
	BKB		
	b		
	CA		
05	CB	55	
	ΣCX		
	CX		
	ΣCY		
	CY		
10	sample number	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 50 USER MODE

ENG ___ FIX 4 SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		2 isotopes	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SEMI-EMPIRICAL NUCLEAR MASS FORMULA

(requires one memory module)

A Semi-Empirical formula is used to calculate approximate binding energies and mass excess for any nucleus with a given nuclear charge, Z, and number of neutrons, N.

Definition: Binding energy (B.E.) = $Z * M_p + N * M_n - M(Z,N)$

M_p = proton mass (energy) in MeV, M_n = neutron mass in MeV

M(Z,N) = mass of nucleus having Z protons and N neutrons.

Mass Excess = M(Z,N) - A * (amu)

A = Z + N, 1 (amu) = M(6,6)/12 --- 1/12 mass of ^{12}C

Weizsacker's Semi-Empirical mass formula contains seven terms

$$M(Z,N) = Z*M_p + N*M_n + E_v + E_s + E_c + E_{sym} + E_{pair}$$

$E_v = -a1 * A$

$E_s = a2 * A^{2/3}$

$E_c = a3 * Z^2/A^{1/3}$

$E_{sym} = a4 * (z-N)^2/A$

$E_{pair} = \pm 34/A^{3/4}$ depending on whether Z and N are both odd or both even.

$E_{pair} = 0$ for odd A nuclei

Notes:

The semiempirical formula has been derived from measured masses and binding energies and is expected to work for nuclei reasonably close to the valley of stability. Usually $N \geq Z$ especially for heavier nuclei.

Example:

What is the binding energy, the mass, mass excess, volume energy, surface energy, coulomb energy, symmetry energy, and pairing energy of the element titanium (Z = 22, N = 26)?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 025	
[XEQ] [ALPHA] NM [ALPHA]	NUM PROT?
22 [R/S]	NUM NEUT?
26 [R/S]	B.E.=-404.5143
[R/S]	B.E./A=-8.4274
[R/S]	M=44,677.9077
[R/S]	M/A=930.7897
[R/S]	M.E.=0.0000
[R/S]	M.E./A=0.0000
[R/S]	EV=-752.6400
[R/S]	EV/A=-15.6800
[R/S]	ES=245.1351
[R/S]	ES/A=5.1070
[R/S]	Ec=95.4884
[R/S]	Ec/A=1.9893
[R/S]	ESYM=9.3667
[R/S]	ESYM/A=0.1951
[R/S]	EP=-1.8644
[R/S]	EP/A=-0.0388

User Instructions

SIZE: 025

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	Initialize program.		[XEQ] NM	NUM PROT?
3.	Input number of protons.	Z	[R/S]	NUM NEUT?
4.	Input number of neutrons	N	[R/S]	B.E.=
	and solve for binding energy terms.		[R/S]	B.E./A=
5.	Calculate the mass of nucleus.		[R/S]	M=
5a.	Calculate the mass per nucleon.		[R/S]	M/A=
6.	Calculate the mass excess of nucleus.		[R/S]	M.E.=
6a.	Calculate mass excess per nucleon.		[R/S]	M.E./A=
7.	Calculate the volume energy.		[R/S]	EV=
7a.	Calculate volume energy per nucleon.		[R/S]	EV/A=
8.	Calculate surface energy.		[R/S]	ES=
8a.	Calculate surface energy per nucleon.		[R/S]	ES/A=
9.	Calculate coulomb energy.		[R/S]	Ec=
9a.	Calculate coulomb energy per nucleon.		[R/S]	Ec/A=
10.	Calculate symmetry energy.		[R/S]	ESYM=
10a.	Calculate symmetry energy per nucleon.		[R/S]	ESYM/A=
11.	Calculate pairing energy.		[R/S]	EP=
11a.	Calculate pairing energy per nucleon.		[R/S]	EP/A=
12.	Reset for new case.		[R/S]	NUM PROT?

Program Listings

Initialize and store constants

```
01◆LBL "NM"
02 CF 01
03 CF 00
04 CLRG
05 -931.504
06 STO 08
07 938.793
08 STO 09
09 939.576
10 STO 10
11 -15.68
12 STO 11
13 18.56
14 STO 12
15 .717
16 STO 13
17 28.1
18 STO 14
19 -17
20 STO 15
```

Prompt for inputs

```
21 "NUM PRO
T?"
22 PROMPT
23 FS?C 22
24 STO 01
25 "NUM NEU
T?"
26 PROMPT
27 FS?C 22
28 STO 02
29 RCL 02
30 RCL 01
31 +
32 STO 00
33 XEQ C
34 XEQ D
35 XEQ E
36 XEQ a
37 XEQ b
38 XEQ c
39 XEQ d
40 XEQ e
41 GTO "NM"
```

Calculate Binding Energy

```
42◆LBL C
43 XEQ 01
44 19.024
45 STO 03
46 XEQ 04
47 "B.E.="
48 ARCL X
49 PROMPT
50 RCL 00
51 /
52 "B.E./A=
"
53 ARCL X
54 PROMPT
55 RTN
```

Calculate mass

```
56◆LBL D
57 17.024
58 STO 03
59 XEQ 04
60 "M="
61 ARCL X
62 PROMPT
63 RCL 00
64 /
65 "M/A="
66 ARCL X
67 PROMPT
68 RTN
```

Calculate mass excess

```
69◆LBL E
70 XEQ 01
71 16.024
72 XEQ 04
73 "M.E.="
74 ARCL X
75 PROMPT
76 RCL 00
77 /
78 "M.E./A=
"
79 ARCL X
80 PROMPT
81 RTN
```

Display volume energy

```
82◆LBL a
83 RCL 20
84 "EV="
85 ARCL X
86 PROMPT
87 RCL 00
88 /
89 "EV/A="
90 ARCL X
91 PROMPT
92 RTN
```

Display surface energy

```
93◆LBL b
94 RCL 21
95 "ES="
96 ARCL X
97 PROMPT
98 RCL 00
```

Program Listings

```
/
"ES/A="
ARCL X
PROMPT
RTN
```

Display Coulomb Energy

```
♦LBL c
RCL 22
"Ec="
ARCL X
PROMPT
RCL 00
/
"Ec/A="
ARCL X
PROMPT
RTN
```

Display Symmetry Energy

```
♦LBL d
RCL 23
"ESYM="
ARCL X
PROMPT
RCL 00
/
"ESYM/A="
ARCL X
PROMPT
RTN
```

Display Pairing Energy

```
♦LBL e
RCL 24
"EP="
ARCL X
PROMPT
RCL 00
/
"EP/A="
ARCL X
PROMPT
RTN
```

Calculation of all terms

```
♦LBL 01
8
STO 03
RCL 00
XEQ 02
RCL 01
XEQ 02
RCL 02
XEQ 02
RCL 00
XEQ 02
RCL 00
3
1/X
Y↑X
X↑2
XEQ 02
RCL 01
X↑2
RCL 00
3
1/X
Y↑X
/
XEQ 02
RCL 01
RCL 02
-
X↑2
RCL 00
/
XEQ 02
-1
RCL 01
Y↑X
-1
RCL 02
Y↑X
+
RCL 00
.75
Y↑X
/
XEQ 02
SF 01
RTN
♦LBL 02
RCL IND 03
*
RCL 03
9
+
STO 03
X<>Y
STO IND 03
8
ST- 03
RTN
♦LBL 04
0
STO 06
♦LBL 05
ISG 03
```

Program Listings

```
200 GTO 10
201 GTO 06
202♦LBL 10
203 RCL IND
03
204 ST+ 06
205 GTO 05
206♦LBL 06
207 RCL 06
208 RTN
209 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	A	50	
	Z		
	N		
	indirect adress		
05		55	
	used		
	-amu		
	Mp		
10	Mn	60	
	$-a_1$		
	a_2		
	a_3		
	a_4		
15	$-a_5$	65	
	Z Mp		
	N Mn		
20	EV	70	
	ES		
	Ec		
	ESYM		
	EP		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 025 TOT. REG. 89 USER MODE

ENG ____ FIX 4 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Used	
01		Used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CLEBSCH-GORDON COEFFICIENTS AND 3J SYMBOLS EVALUATION

This program will evaluate all valid Clebsch-Gordon Coefficients and/or "3j" symbols coupling two states of angular momentum which are small enough so that the capacity of the calculator's factorial function is not exceeded. The fundamental formula used by the program is the Racah Formula:

$$\underbrace{\begin{pmatrix} j_1 & j_2 & J \\ m_1 & m_2 & -M \end{pmatrix}}_{\text{3j Symbol}} =$$

$$(-1)^{j_1-j_2+M}\sqrt{\Delta(j_1 j_2 J)}\sqrt{(j_1+m_1)!(j_1-m_1)!(j_2+m_2)!(j_2-m_2)!(J+M)!(J-M)!}$$

$$\cdot \sum_t (-1)^t \Big[t!(J-j_2+t+m_1)!(J-j_1+t-m_2)!(j_1+j_2-J-t)!$$

$$\cdot (j_1-t-m_1)!(j_2-t+m_2)! \Big]^{-1}$$

with $\Delta(j_1 j_2 J) = \Big[(j_1+j_2-J)!(j_2+J-j_1)!(J+j_1-j_2)!\Big] \div (j_1+j_2+J+1)!$

subject to the restraints 1) $|j_1 - j_2| \leq J \leq |j_1 + j_2|$

2) $m_1 + m_2 = M$

The Clebsch-Gordon Coefficient, $\langle j_1 j_2 m_1 m_2 | JM \rangle = \dfrac{\sqrt{2J+1}}{(-1)^{j_1-j_2+m}} \begin{pmatrix} j_1 & j_2 & J \\ m_1 & m_2 & -M \end{pmatrix}$

If any one term in the Racah formula is greater than 69, an out of range message will result. If illegitimate values are entered for j_1, j_2, and J or m_1, m_2, and M spurious results (i.e., non-zero) may be obtained <u>or</u> the calculator may get caught in a "loop" which will not terminate until the "t" value in the Racah formula exceeds 69.

Example:

Suppose the C.G. coefficient $\langle j_1 j_2 m_1 m_2 | JM \rangle$ is needed with $j_1 = 3/2$, $j_2 = 2$, $J = 5/2$, $M_1 = 1/2$, $M_2 = 0$, $M = 1/2$

Keystrokes:	Display:
[XEQ] ALPHA] SIZE [ALPHA] 018	
[XEQ] [ALPHA] CGC [ALPHA]	J1=?
1.5 [R/S]	J2=?
2 [R/S]	J=?
2.5 [R/S]	M1=?
.5 [R/S]	M2=?
0 [R/S]	M=?
.5 [R/S]	C.G.=2.9277 E-1
[R/S]	3J=1.1952 E-1

User Instructions

SIZE: 018

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] CGC	J1=?
3.	Key in J1 value.	J1	[R/S]	J2=?
4.	Key in J2 value.	J2	[R/S]	J=?
5.	Key in J value.	J	[R/S]	M1=?
6.	Key in M1 value.	M1	[R/S]	M2=?
7.	Key in M2 value.	M2	[R/S]	M=?
8.	Key in M.	M	[R/S]	C.G.=
			[R/S]	3J=

Program Listings

```
01*LBL "CGC"
02 SCI 4
03 CLRG
04 "J1=?"
05 PROMPT
06 STO 01
07 "J2=?"
08 PROMPT
09 STO 02
10 "J=?"
11 PROMPT
12 STO 03
13 "M1=?"
14 PROMPT
15 STO 11
16 "M2=?"
17 PROMPT
18 STO 12
19 "M=?"
20 PROMPT
21 CHS
22 STO 10
```

Clears Registers and stores variables

```
23 RCL 01
24 RCL 02
25 RCL 03
26 -
27 +
28 FACT
29 STO 04
30 RCL 02
31 RCL 03
32 RCL 01
33 -
34 +
35 FACT
36 ST* 04
37 RCL 03
38 RCL 01
39 RCL 02
40 -
41 +
42 FACT
43 ST* 04
44 1
45 RCL 01
46 RCL 02
47 RCL 03
48 +
49 +
50 +
51 FACT
52 RCL 04
53 X<>Y
54 /
55 SQRT
56 STO 04
```

Begins computation

STORES $\sqrt{\Delta\ (J1\ J2\ J)}$

```
57 RCL 01
58 RCL 11
59 +
60 FACT
61 RCL 01
62 RCL 11
63 -
64 FACT
65 *
66 STO 05
67 RCL 02
68 RCL 12
69 +
70 FACT
71 RCL 02
72 RCL 12
73 -
74 FACT
75 *
76 ST* 05
77 RCL 03
78 RCL 10
79 +
80 FACT
81 RCL 03
82 RCL 10
83 -
84 FACT
85 *
86 ST* 05
87 RCL 05
88 SQRT
89 STO 05
90 RCL 01
91 RCL 02
92 RCL 10
93 +
94 -
95 -1
96 X<>Y
97 Y↑X
98 STO 06
99 RCL 04
100 RCL 05
101 RCL 06
```

Program Listings

102 *		153 -	
103 *	Computes summa-	154 X<0?	
104 STO 13	tion in Racah	155 GTO 03	
105 CLX	Formula	156 FACT	
106 SF 00		157 STO 07	
107 RCL 11		158 RCL 02	
108 STO 15		159 RCL 16	
109 RCL 12		160 +	
110 STO 16		161 RCL 14	
111 RCL 10		162 -	
112 STO 17		163 X<0?	
113 0		164 GTO 03	
114 STO 14		165 FACT	
115♦LBL 02		166 STO 08	
116 RCL 03		167 CF 00	
117 RCL 14		168 RCL 04	
118 RCL 15		169 RCL 05	
119 +		170 RCL 06	
120 +		171 *	
121 RCL 02		172 *	
122 -		173 RCL 07	
123 X<0?		174 RCL 08	
124 GTO 03		175 *	
125 FACT		176 *	
126 STO 04		177 RCL 14	
127 RCL 03		178 FACT	
128 RCL 14		179 *	
129 +		180 1/X	
130 RCL 01		181 -1	
131 RCL 16		182 RCL 14	
132 +		183 Y↑X	
133 -		184 *	
134 X<0?		185 ST+ 09	
135 GTO 03		186 1	
136 FACT		187 ST+ 14	
137 STO 05		188 GTO 02	
138 RCL 01		189♦LBL 03	Checks to see if
139 RCL 02		190 1	Σ finished
140 +		191 ST+ 14	t
141 RCL 03		192 FS? 00	
142 RCL 14		193 GTO 02	
143 +		194 RCL 09	
144 -		195 RCL 13	
145 X<0?		196 *	
146 GTO 03		197 STO 00	Computes Clebsch-
147 FACT		198 -1	Gordon Coeffi-
148 STO 06		199 RCL 01	cient
149 RCL 01		200 RCL 17	
150 RCL 14		201 -	
151 RCL 15		202 RCL 02	
152 +		203 -	

Program Listings

```
204 Y↑X
205 RCL 03
206 2
207 *
208 1
209 +
210 SQRT
211 *
212 RCL 00
213 *
214 "C.G.="
215 ARCL X
216 AVIEW
217 STOP
218 "3J="
219 ARCL 00
220 AVIEW
221 RTN
222 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	3J Value	50	
	J1		
	J2		
	J		
	Used		
05	Used	55	
	Used		
	Used		
	Used		
	Used		
10	-M	60	
	M1		
	M2		
	Used		
	Used		
15	Used	65	
	Used		
	Used		
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 018 TOT. REG. 59 USER MODE
ENG ____ FIX 4 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	Calculation not finished	Calculation finished

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

32-P REMAINING ON DAY OF YEAR

This program calculates decays per minute (DPM) and counts per minute (CPM) remaining on any day given millicuries (mCi) on an earlier day. This program also calculates CPM and mCi remaining of 32-P sample on a day given CPM on an earlier date.

$$\text{mCi on date 2} = (\text{initial mCi})(0.5)^n$$

where $n = 14.3/\Delta\text{days}$ = the number of half lives of 32P which have occurred.

$$\text{DPM on date 2} = (\text{mCi on date 2})(2.2 \times 10^9 \text{ DPM})$$

$$\text{CPM on date 2} = 0.3 \text{ DPM}$$

(assumes 30% counting efficiency and no quenching)

Note:

Program fails if Date 1 = Date 2 or if Date 1 is more recent than Date 2.

Reference:

HP-67/HP-97 USERS LIBRARY Physics Solutions Book.

Example:

1) How many DPM and CPM remain of a 0.130 mCi 32P sample given: date of specific radioactivity rating as Feb. 1, 1979, and present date Aug. 4, 1979.

2) How many CPM and mCi remain on Aug. 4, 1979, of a 4.2×10^6 CPM sample of 32P as measured on Aug. 11, 1978?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 013	
[XEQ] [ALPHA] 32P [ALPHA]	DATE1?
2.011979 [R/S]	DATE2?
8.041979 [R/S]	MCI ON 1?
0.130 [R/S]	DPM=3.8280E4
[R/S]	CPM=1.148E4
[R/S]	DATE 1?
8.111978 [R/S]	DATE 2?
8.041979 [R/S]	MCI ON 1?
[R/S]	CPM ON 1?
4.2 [EEX]6 [R/S]	CPM=1.E-1
[R/S]	MCI=1.85E-10

User Instructions

SIZE: 013

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize the program.		[XEQ] 32D	DATE1?
3.	Input first date.	MM.DDYYYY	[R/S]	DATE2?
4.	Input second (more recent) date.	MM.DDYYYY	[R/S]	MCI ON 1?
5.	If mCi on date 1 is unknown, go to step 9.			
6.	Input mCi on date 1 and calculate DPM			
	on date 2.	mCi	[R/S]	DPM =
7.	Calculate CPM on date 2		[R/S]	CPM =
8.	To reset for new case.		[R/S]	DATE1?
9.	If CPM on date 1 is known, press [R/S].		[R/S]	CPM ON 1?
10.	Input CPM on date 1, and calculate CPM			
	on date 2.	CPM	[R/S]	CPM =
11.	Calculate mCi on date 2.		[R/S]	MCI =
12.	Reset for new case.		[R/S]	DATE1?

Program Listings

Initialize and input prompts

```
 01♦LBL "32P
"
 02 CLRG
 03 CF 22
 04 "DATE1?"
 05 PROMPT
 06 XEQ A
 07 "DATE2?"
 08 PROMPT
 09 XEQ B
 10 XEQ C
 11 CF 22
 12 "MCI ON
1?"
 13 PROMPT
 14 FC?C 22
 15 GTO 05
 16 XEQ D
 17 XEQ E
 18 GTO "32P
"
 19♦LBL 05
 20 "CPM ON
1?"
 21 PROMPT
 22 XEQ D
 23 XEQ c
 24 GTO "32P
"
```

Calculation of Δ Days and storage of constants

```
 25♦LBL A
 26 FIX 2
 27 RCL 04
 28 RCL 02
 29 -
 30 3
 31 GTO 00
 32♦LBL B
 33 RCL 03
 34 RCL 02
 35 +
 36 4
 37♦LBL 00
 38 STO 00
 39 RDN
 40 365.25
 41 STO 05
 42 30.6001
 43 STO 06
 44 RDN
 45 RDN
 46 FS?C 22
 47 GTO 01
 48 STO IND
00
 49 122.1
 50 -
 51 RCL 05
 52 /
 53 INT
 54 STO 09
 55 RCL 05
 56 *
 57 INT
 58 RCL IND
00
 59 -
 60 CHS
 61 STO 01
 62 RCL 06
 63 /
 64 INT
 65 STO 07
 66 RCL 01
 67 X<>Y
 68 RCL 06
 69 *
 70 INT
 71 -
 72 STO 06
 73 RCL 07
 74 1
 75 RCL 08
 76 %
 77 -
 78 -
 79 RCL 07
 80 14
 81 /
 82 XEQ 02
 83 RCL 09
 84 1 E6
 85 *
 86 +
 87 FIX 6
 88 RTN
```

Break date into months, days, and years

```
 89♦LBL 01
 90 RDN
 91 ENTER↑
 92 INT
 93 STO 07
 94 -
 95 1 E2
```

Program Listings

```
 96 *
 97 ENTER↑
 98 INT
 99 STO 08
100 -
101 1 E4
102 *
103 STO 09
104 RCL 07
105 1
106 +
107 ENTER↑
108 1/X
109 .7
110 +
111 CHS
112 XEQ 02
```

Compute Julian day number

```
113 RCL 06
114 *
115 INT
116 RCL 09
117 RCL 05
118 *
119 INT
120 +
121 RCL 08
122 +
123 STO IND
00
124 1720982
125 +
126 FIX 0
127 RTN
128♦LBL 02
129 INT
130 ST+ 09
131 12
132 *
133 -
134 RTN
```

Store Δ Days

```
135♦LBL C
136 FIX 0
137 STO 02
138 FS?C 22
139 RTN
140 RCL 04
141 RCL 03
142 -
143 STO 02
144 RTN
```

Store Constants

```
145♦LBL D
146 STO 10
147 14.3
148 STO 11
149 2.2 E9
150 STO 12
151 RTN
```

Compute DPM and CPM

```
152♦LBL E
153 2
154 RCL 02
155 RCL 11
156 /
157 Y↑X
158 1/X
159 RCL 10
160 RCL 12
161 *
162 *
163 SCI 4
164 "DPM="
165 ARCL X
166 PROMPT
167 ENTER↑
168 .3
169 *
170 "CPM="
171 ARCL X
172 PROMPT
173 RTN
```

Compute CPM and mCi

```
174♦LBL c
175 RCL 02
176 RCL 11
177 /
178 .5
179 X<>Y
180 Y↑X
181 RCL 10
182 *
183 "CPM="
184 ARCL X
185 PROMPT
186 .3
187 /
188 RCL 12
189 /
190 SCI 2
191 "MCI="
192 ARCL X
193 PROMPT
194 RTN
195 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	indirect address	50	
	used		
	Δ days		
	Day 1		
	Day 2		
05	365.25	55	
	30.6001		
	MM		
	DD		
	YYYY		
10	used	60	
	used		
	used		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 013 TOT. REG. 61 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Physics

PHYSICS

BLACK BODY THERMAL RADIATION

PROGRAM REGISTERS NEEDED: 48

HEWLETT PACKARD
SOLUTION BOOK:
PHYSICS

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (10 - 13)

ROW 4 (13 - 17)

ROW 5 (17 - 20)

ROW 6 (20 - 23)

ROW 7 (23 - 27)

ROW 8 (27 - 30)

ROW 9 (30 - 33)

ROW 10 (33 - 36)

ROW 11 (36 - 42)

ROW 12 (42 - 49)

ROW 13 (49 - 57)

ROW 14 (58 - 62)

ROW 15 (63 - 75)

ROW 16 (76 - 83)

ROW 17 (83 - 95)

ROW 18 (96 - 108)

ROW 19 (109 - 121)

ROW 20 (122 - 131)

ROW 21 (131 - 141)

ROW 22 (142 - 146)

ROW 23 (147 - 149)

ROW 24 (149 - 158)

ROW 25 (159 - 167)

ROW 26 (167 - 170)

BLACK HOLE CHARACTERISTICS

PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTION BOOK:
PHYSICS

ROW 1 (1 - 3)

ROW 2 (3 - 6)

ROW 3 (7 - 12)

ROW 4 (12 - 20)

ROW 5 (21 - 21)

ROW 6 (21 - 28)

ROW 7 (29 - 32)

ROW 8 (32 - 39)

ROW 9 (39 - 45)

ROW 10 (45 - 52)

ROW 11 (53 - 60)

ROW 12 (60 - 67)

ROW 13 (67 - 74)

ROW 14 (75 - 81)

SPECIAL RELATIVITY CONVERSIONS

PROGRAM REGISTERS NEEDED: 41

HEWLETT PACKARD
SOLUTION BOOK:
PHYSICS

ROW 1 (1 – 5)

ROW 2 (5 – 11)

ROW 3 (12 – 12)

ROW 4 (13 – 21)

ROW 5 (22 – 30)

ROW 6 (30 – 36)

ROW 7 (37 – 48)

ROW 8 (49 – 56)

ROW 9 (57 – 64)

ROW 10 (65 – 72)

ROW 11 (73 – 81)

ROW 12 (82 – 90)

ROW 13 (90 – 99)

ROW 14 (100 – 106)

ROW 15 (107 – 113)

ROW 16 (113 – 121)

ROW 17 (121 – 127)

ROW 18 (128 – 135)

ROW 19 (135 - 145)

ROW 20 (146 - 153)

ROW 21 (154 - 161)

ROW 22 (162 - 167)

ROW 1 (1 - 2)

ROW 2 (2 - 6)

ROW 3 (7 - 13)

ROW 4 (13 - 18)

ROW 5 (18 - 24)

ROW 6 (25 - 28)

ROW 7 (28 - 31)

ROW 8 (31 - 33)

ROW 9 (33 - 37)

ROW 10 (38 - 50)

ROW 11 (51 - 62)

ROW 12 (63 - 71)

ROW 13 (72 - 79)

ROW 14 (80 - 89)

ROW 15 (90 - 95)

ROW 16 (96 - 101)

ROW 17 (102 - 107)

ROW 18 (107 - 116)

ROW 19 (117 - 128)

ROW 20 (128 - 131)

ROW 21 (131 - 140)

ROW 22 (141 - 149)

ROW 23 (150 - 158)

ROW 24 (159 - 169)

ROW 25 (169 - 179)

ROW 26 (179 - 189)

ROW 27 (190 - 199)

ROW 28 (199 - 203)

EINSTEINS TWIN PARADOX

PROGRAM REGISTERS NEEDED: 16

HEWLETT PACKARD
SOLUTION BOOK:
PHYSICS

ROW 1 (1 - 4)

ROW 2 (4 - 9)

ROW 3 (10 - 17)

ROW 4 (17 - 24)

ROW 5 (24 - 31)

ROW 6 (31 - 37)

ROW 7 (38 - 46)

ROW 8 (47 - 52)

ROW 9 (53 - 56)

DELTA–V ORBIT SIMULATOR

HEWLETT PACKARD
SOLUTION BOOK:
PHYSICS

PROGRAM REGISTERS NEEDED: 67

ROW 1 (1 – 3)

ROW 2 (4 – 8)

ROW 3 (8 – 11)

ROW 4 (12 – 14)

ROW 5 (14 – 20)

ROW 6 (21 – 27)

ROW 7 (28 – 33)

ROW 8 (34 – 39)

ROW 9 (40 – 47)

ROW 10 (48 – 54)

ROW 11 (55 – 65)

ROW 12 (66 – 71)

ROW 13 (72 – 79)

ROW 14 (79 – 84)

ROW 15 (85 – 94)

ROW 16 (95 – 102)

ROW 17 (103 – 112)

ROW 18 (113 – 116)

ROW 19 (117 – 120)

ROW 20 (120 – 128)

ROW 21 (129 – 134)

ROW 22 (135 – 147)

ROW 23 (148 – 160)

ROW 24 (161 – 172)

ROW 25 (173 – 182)

ROW 26 (182 – 191)

ROW 27 (192 – 199)

ROW 28 (200 – 206)

ROW 29 (207 – 212)

ROW 30 (213 – 223)

ROW 31 (223 – 233)

ROW 32 (234 – 246)

ROW 33 (247 – 259)

ROW 34 (260 – 271)

ROW 35 (272 – 284)

ROW 36 (284 – 293)

ROW 37 (293 - 293)

PROGRAM REGISTERS NEEDED: 47

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (10 - 16)

ROW 4 (17 - 23)

ROW 5 (23 - 30)

ROW 6 (30 - 37)

ROW 7 (38 - 49)

ROW 8 (49 - 59)

ROW 9 (60 - 68)

ROW 10 (69 - 79)

ROW 11 (80 - 91)

ROW 12 (91 - 99)

ROW 13 (100 - 109)

ROW 14 (109 - 121)

ROW 15 (122 - 132)

ROW 16 (133 - 141)

ROW 17 (142 - 154)

ROW 18 (155 - 163)

ROW 19 (164 - 175)

ROW 20 (176 - 186)

ROW 21 (187 - 198)

ROW 22 (199 - 208)

ROW 23 (209 - 216)

ROW 24 (216 - 222)

ROW 25 (223 - 228)

ISOTOPE OVERLAP CORRECTIONS

PROGRAM REGISTERS NEEDED: 40

HEWLETT PACKARD
SOLUTION BOOK:
PHYSICS

ROW 1 (1 - 5)

ROW 2 (5 - 9)

ROW 3 (10 - 11)

ROW 4 (11 - 17)

ROW 5 (17 - 19)

ROW 6 (19 - 24)

ROW 7 (24 - 29)

ROW 8 (30 - 34)

ROW 9 (34 - 38)

ROW 10 (39 - 47)

ROW 11 (48 - 59)

ROW 12 (60 - 70)

ROW 13 (71 - 83)

ROW 14 (84 - 90)

ROW 15 (91 - 96)

ROW 16 (97 - 102)

ROW 17 (103 - 107)

ROW 18 (108 - 111)

ROW 19 (112 - 116)

ROW 20 (116 - 123)

ROW 21 (123 - 133)

ROW 22 (133 - 133)

ROW 1 (1 - 5)

ROW 2 (5 - 7)

ROW 3 (7 - 11)

ROW 4 (11 - 15)

ROW 5 (15 - 21)

ROW 6 (21 - 25)

ROW 7 (25 - 28)

ROW 8 (29 - 35)

ROW 9 (36 - 40)

ROW 10 (40 - 44)

ROW 11 (44 - 47)

ROW 12 (47 - 52)

ROW 13 (52 - 57)

ROW 14 (58 - 65)

ROW 15 (65 - 70)

ROW 16 (71 - 73)

ROW 17 (73 - 78)

ROW 18 (78 - 84)

ROW 19 (84 - 90)

ROW 20 (90 - 96)

ROW 21 (97 - 103)

ROW 22 (104 - 110)

ROW 23 (111 - 116)

ROW 24 (116 - 122)

ROW 25 (122 - 126)

ROW 26 (127 - 133)

ROW 27 (133 - 141)

ROW 28 (141 - 147)

ROW 29 (147 - 157)

ROW 30 (158 - 168)

ROW 31 (168 - 177)

ROW 32 (177 - 184)

ROW 33 (185 - 195)

ROW 34 (196 - 204)

ROW 35 (204 - 209)

ROW 1 (1 – 4)

ROW 2 (4 – 10)

ROW 3 (10 – 16)

ROW 4 (16 – 23)

ROW 5 (24 – 36)

ROW 6 (36 – 47)

ROW 7 (48 – 60)

ROW 8 (61 – 73)

ROW 9 (74 – 85)

ROW 10 (86 – 96)

ROW 11 (97 – 108)

ROW 12 (109 – 119)

ROW 13 (120 – 131)

ROW 14 (131 – 142)

ROW 15 (143 – 154)

ROW 16 (155 – 164)

ROW 17 (165 – 176)

ROW 18 (177 – 187)

ROW 19 (187 – 195)

ROW 20 (196 – 206)

ROW 21 (207 – 214)

ROW 22 (215 – 222)

ROW 23 (222 – 222)

32-P REMAINING ON DAY OF YEAR

PROGRAM REGISTERS NEEDED: 49

ROW 1 (1 - 4)

ROW 2 (4 - 7)

ROW 3 (7 - 12)

ROW 4 (12 - 15)

ROW 5 (16 - 20)

ROW 6 (20 - 22)

ROW 7 (23 - 27)

ROW 8 (28 - 38)

ROW 9 (39 - 42)

ROW 10 (42 - 49)

ROW 11 (49 - 58)

ROW 12 (59 - 71)

ROW 13 (72 - 82)

ROW 14 (82 - 91)

ROW 15 (92 - 101)

ROW 16 (101 - 112)

ROW 17 (112 - 123)

ROW 18 (123 - 128)

ROW 19 (129 - 137)

ROW 20 (138 - 147)

ROW 21 (147 - 153)

ROW 22 (154 - 164)

ROW 23 (164 - 170)

ROW 24 (170 - 178)

ROW 25 (179 - 186)

ROW 26 (186 - 192)

ROW 27 (193 - 195)

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

* Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

PHYSICS

BLACK BODY THERMAL RADIATION
BLACK HOLE CHARACTERISTICS
SPECIAL RELATIVITY CONVERSIONS
THREE DIMENSIONAL SPECIAL RELATIVITY
EINSTEIN'S TWIN PARADOX
DELTA-V-ORBIT SIMULATOR
EQUATIONS OF MOTION
ISOTOPE OVERLAP CORRECTIONS
SEMI-EMPIRICAL NUCLEAR MASS FORMULA
CLEBSCH-GORDON COEFFICIENTS AND 3J SYMBOLS EVALUATION
32-P REMAINING ON DAY OF YEAR

HEWLETT PACKARD

00041-90142 Printed in U.S.A.